National Accounting Review

NAR, 7(4): 602–629.
DOI: 10.3934/NAR.2025025
Received: 24 July 2025
Revised: 22 September 2025
Accepted: 24 October 2025
Published: 09 December 2025

https://www.aimspress.com/journal/NAR

Research article

Linking policy targets in navigating through fiscal-monetary crosswinds

Serhii Shvets*

Department of Modeling and Forecasting of Economic Development, State Organization "Institute for Economics and Forecasting of the National Academy of Sciences of Ukraine", 26, Panasa Myrnoho, Kyiv, 01011, Ukraine

* **Correspondence:** Email: smserg@ukr.net.

Supplementary

Appendix A. Dynare code of the DSGE model

```
//NK model: Linking policy targets in navigating through fiscal-monetary crosswinds

var Y G IG CG IP CP CR CNR KG KP LR LNR L T W M
B Bint Bext BY NX P Pext PI PIW i iext r s MC;

varexo nu;

parameters beta h altphi chiL chiM phi delta alphaK omega omegaW
thetaW theta eta alphaG epsilon0 epsilon1 upsilon sigmaG
sigmaB rhoi rhoPI rhoY rhos rhoB rhoBext rhoPext kd kappaG;

//Calibration
beta = 0.9828;
h = 0.7;
altphi = 0.3;
chiM = 0.4;
chiL = 0.3;
```

```
phi = 2;
delta = 0.025;
alphaK = 1/3;
omega = 8;
omegaW = 20;
theta = 0.75;
thetaW = 0.75;
eta = 0.6;
//Fiscal policy parameters
alphaG = 0.15;
epsilon0 = 0.5;
epsilon1 = 0.1;
kd = 0.9;
sigmaG = 0.3;
sigmaB = 0.1;
//Taylor's rule parameters
rhoi = 0.6;
rhoPI = 1.5;
rhoY = 0.1;
rhos = 0.5;
rhoB = 0.1;
//Foreign interest rate
rhoBext = 0.01;
rhoPext = -0.05;
//Steady-state levels
upsilon = 0.08;
//Autoregressive shock parameters
kappaG = 0.6;

model(linear);
//computation of the steady state
#iss = 1/beta-1;
#iextss = iss;
#rss = iss+delta;
#LRss = (beta/chiL)^(1/(phi+1));
#LNRss = LRss;
#Lss = LRss;
#MCss = (1-beta*theta)*(omega-1)/omega;
#CGss = (MCss*(alphaK/rss)^alphaK*((1-alphaK)*(1-chiL*LNRss^(phi+1))/
(altphi*chiL*LNRss^phi))^(1-alphaK)*(upsilon/(1-upsilon)*(epsilon0/delta)^
(1/(1-epsilon1)))^alphaG)^(1/(1-alphaK-alphaG));
#Gss = CGss/(1-upsilon);
#IGss = Gss-CGss;
#CRss = beta/(1-beta)*altphi*CGss;
```

```
#CNRss = CRss;
#CPss = CRss;
#Wss = (Gss+CRss)*rss/Lss/(alphaK/(1-alphaK)*iss+rss);
#Mss = chiM/chiL*(iss+1)/iss*Wss/LRss^phi;
#KPss = alphaK/(1-alphaK)*Wss*Lss/rss;
#KGss = (epsilon0/delta)^(1/(1-epsilon1))*IGss;
#Yss = KPss^alphaK*Lss^(1-alphaK)*KGss^alphaG;
#IPss = delta*KPss;
#NXss = Yss-CPss-IPss-Gss;
#Bextss = NXss/iextss;
#Bintss = 2*Mss/kd-Bextss;
#Tss = Gss+iss*Bintss+NXss;

//Log-linearized equations
//1-Phillips equation for Ricardian wage
PIW = beta*PIW(+1)+(1-thetaW)*(1-beta*thetaW)/thetaW*(1/((1-beta*h)*(CRss*(1-h)+
altphi*CGss))*((1-beta*h)*(CRss*(CR+CR(+1))-h*CRss*(CR(-1)+CR)+altphi*CGss*
(CG+CG(+1)))-CRss*(CR(+1)-beta*h*CR)+h*CRss*(CR-beta*h*CR(-1))-altphi*CGss*
(CG(+1)-beta*h*CG))+phi*LR-(W-P));
//2-Wage inflation
PIW = W-W(-1);
//3-Ricardian household budget constraint
CRss*CR+KPss*(KP-KP(-1)*(rss+1-delta)-rss*r)+Mss*(M-M(-1))+Bintss*(Bint-Bint(-1)*
(1+iss)-iss*(i(-1)-P))+Bextss*(Bext-Bext(-1)*(1+iextss)-iextss*(s+iext(-1)-P)) =
Wss*LRss*(W-P+LR)-Tss*T;
//4-Ricardian Euler equation
CRss*(CR(+1)-h*CR)+altphi*CGss*CG(+1)-beta*h*(CRss*(CR-h*CR(-1))+
altphi*CGss*CG) = (1-beta*h)*(CRss*(1-h)+altphi*CGss)*(phi*LR-W+P);
//5-Labor supply of Ricardian household
phi*LR = W-M-i/(iss+1);
//6-Phillips equation for non-Ricardian wage
PIW = beta*PIW(+1)+(1-thetaW)*(1-beta*thetaW)/thetaW*((CNRss*CNR+altphi*
CGss*CG)/(CNRss+altphi*CGss)+phi*LNR-W+P);
//7-Non-Ricardian household budget constraint
CNR = LNR+W-P;
//8-Labor supply of non-Ricardian household
CNRss*(CNR+phi*LNR)+altphi*CGss*(CG+phi*LNR) = Wss/(chiL*LNRss^phi)*(W-P);
//9-Investment-capital trade-off
iss*i = rss*r(+1)+(rss+1-delta)*PI(+1);
//10-Law of motion of private capital
KP = (1-delta)*KP(-1)+delta*IP;
//11-Production function
Y = alphaK*KP(-1)+(1-alphaK)*L+alphaG*KG(-1);
//12-Marginal cost
```

```
MC = (1-alphaK)*(W-P)+alphaK*r-alphaG*KG(-1);
//13-Labor-capital trafe-off
KP(-1)+r = W-P+L;
//14-Phillips equation
PI = beta*PI(+1)+(1-theta)*(1-beta*theta)/theta*(MC-P);
//15-Inflation
PI(+1) = P(+1)-P;
//16-Nominal exchange rate
s = P-Pext;
//17-Money supply
M = (Bintss*Bint+Bextss*(s+Bext))/(Bintss+Bextss)+(Bintss*(Bint-P)+
Bextss*(s+Bext-P))/Yss;
//18-Government budget constraint
Bintss*(Bint-Bint(-1))+Bextss*(Bext-Bext(-1))+Mss*(M-M(-1))+Tss*T =
Gss*G+Bintss*iss*(Bint(-1)+i(-1)-P)+Bextss*iextss*(Bext(-1)+s+iext(-1)-P);
//19-Fiscal policy rule
//Tss*T = sigmaG*Gss*G+sigmaB*(Bintss*(Bint(-1)-P(-1))+Bextss*(s(-1)+Bext(-1)-P(-1)));
Tss*T = sigmaG*Gss*G+sigmaB*Yss*(M(-1)-(Bintss*Bint(-1)+Bextss*(s(-1)+Bext(-1)))/
(Bintss+Bextss));
//20-Law of motion of public capital
KG = KG(-1)*(1-delta+epsilon1*delta)+delta*(IG-epsilon1*IG(-1));
//21-Foreign interest rate
iextss*iext = s(+1)-s+rhoBext*Bextss*Bext+rhoPext*Pext;
//22-Uncovered interest parity
i-iext = s(+1)-s;
//23-Tailor rule
//iss*i = rhoi*iss*i(-1)+rhoPI*PI(-1)+rhoY*Yss*Y(-1)+rhos*s(-1)+
//rhoB*(Bintss*(Bint(-1)-P(-1))+Bextss*(s(-1)+Bext(-1)-P(-1)))/Yss;
iss*i = rhoi*iss*i(-1)+rhoPI*PI(-1)+rhoY*Yss*Y(-1)+rhos*s(-1)+
rhoB*(M(-1)-(Bintss*Bint(-1)+Bextss*(s(-1)+Bext(-1)))/(Bintss+Bextss));
//24-External account
NXss*(NX+Pext)+Bextss*(Bext-Bext(-1)) = iextss*Bextss*(iext(-1)+Bext(-1));
//25-Agregate consumption
CP = (1-eta)*CR+eta*CNR;
//26-Public spending
G = upsilon*IG+(1-upsilon)*CG;
//27-Equilibrium condition
Yss*Y = CPss*CP+IPss*IP+Gss*G+NXss*NX;
//28-Debt total
(Bintss+Bextss)*B = Bintss*Bint+Bextss*(s+Bext);
//29-Debt total ratio
BY = Bint+Bext+s-2*Y-2*P;
//30-Public spending shock
G = kappaG*G(-1)+nu;
```

```
end;

steady;
check(qz_zero_threshold=1e-20);

shocks;
var nu;
stderr 1;
end;

stoch_simul(order=1, periods=1000, irf=40, qz_zero_threshold=1e-20)
Y T KG IP CNR CR B BY PI i s;

varobs Y T KG IP CNR CR B BY PI i s;

estimated_params;
eta, gamma_pdf,1.0,0.1;
kd, gamma_pdf, 2.0,0.15;
sigmaB, beta_pdf,0.25,0.01;
rhoB, beta_pdf,0.25,0.01;
end;

dynare_sensitivity(identification=1, morris=2, Nsam=512);
```